Pensieve header: Analyzing Reidemeister moves by hand.

Integration

As in Talks/Beijing-2407

In[∘]:=
```
CF[ω_. ℰ_𝔼] := CF[ω] CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[q_] :=
  Expand@Collect[q, ϵ | (p | x | π | ξ)_ | (p | x | π | ξ)₊ | (p | x | π | ξ)₋, F] /. F → Factor;
```

In[∘]:=
```
𝔼 /: 𝔼[A_] 𝔼[B_] := 𝔼[A + B];
```

In[∘]:=
```
(ω1_. 𝔼[L1_]) ≡ (ω2_. 𝔼[L2_]) := Simplify[(ω1 == ω2) ∧ (L1 == L2)]
```

In[∘]:=
```
$π = Identity;
```

In[∘]:=
```
Unprotect[Integrate];
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, L0, Q, Δ, G, Z0, Z, λ, DZ, DDZ, FZ, a, b},
   n = Length@vs; L0 = L /. ϵ → 0;
   Q = Table[(-∂_{vs⟦a⟧,vs⟦b⟧} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = Det[Q]) == 0, Return@"Degenerate Q!"];
   Z = Z0 = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q];
   FixedPoint[(DZ = Table[∂_v Z, {v, vs}];
      DDZ = Table[∂_u DZ, {u, vs}];
      FZ = Sum[G⟦a, b⟧ (DDZ⟦a, b⟧ + DZ⟦a⟧ DZ⟦b⟧), {a, n}, {b, n}] / 2;
      Z = CF[Z0 + ∫_0^λ $π[FZ] ⅆλ]) &, Z];
   PowerExpand@Factor[ω Δ^(-1/2)] 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]];
Protect[Integrate];
```

In[∘]:=
```
∫𝔼[-μ x^2 / 2 + ⅈ ξ x] ⅆ{x}
```

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{\xi^2}{2\mu}\right]}{\sqrt{\mu}}$$

In[∘]:=
```
FofG = ∫𝔼[-μ (x - a)^2 / 2 + ⅈ ξ x] ⅆ{x}
```

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{i\,(2\,a\,\mu+i\,\xi)\,\xi}{2\,\mu}\right]}{\sqrt{\mu}}$$

In[∘]:=
$$\int \text{FofG}\,\mathbb{E}[-i\,\xi\,x]\,d\{\xi\}$$

Out[∘]=
$$\mathbb{E}\left[-\frac{1}{2}\,(a-x)^2\,\mu\right]$$

In[∘]:=
$$L = -\frac{1}{2}\,\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\};$$
$$Z12 = \int \mathbb{E}[L]\,d\{x_1, x_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2}{2\,(-b^2+a\,c)} + \frac{b\,\xi_1\,\xi_2}{b^2-a\,c} + \frac{a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{\sqrt{-b^2+a\,c}}$$

In[∘]:=
$$\left\{Z1 = \int \mathbb{E}[L]\,d\{x_1\},\ Z12 == \int Z1\,d\{x_2\}\right\}$$

Out[∘]=
$$\left\{\frac{\mathbb{E}\left[-\frac{(-b^2+a\,c)\,x_2^2}{2\,a} - \frac{b\,x_2\,\xi_1}{a} + \frac{\xi_1^2}{2\,a} + x_2\,\xi_2\right]}{\sqrt{a}},\ \text{True}\right\}$$

In[∘]:=
$$\$\pi = \text{Normal}\left[\# + O[\epsilon]^{13}\right]\ \&;\ \int \mathbb{E}\left[-\phi^2/2 + \epsilon\,\phi^3/6\right]\,d\{\phi\}$$

Out[∘]=
$$\mathbb{E}\left[\frac{5\,\epsilon^2}{24} + \frac{5\,\epsilon^4}{16} + \frac{1105\,\epsilon^6}{1152} + \frac{565\,\epsilon^8}{128} + \frac{82\,825\,\epsilon^{10}}{3072} + \frac{19\,675\,\epsilon^{12}}{96}\right]$$

The Quadratic

Matrices from SSAlexander4Tangles.nb

In[∘]:=



$$\left(\begin{array}{l} Q_{i\_,j\_,k\_,l\_} := p_{\_}\,x_{\_}\ /.\ \{p_{\_} \to p_i - p_k,\ x_{\_} \to x_i - x_k\}; \\ \bar{Q}_{i\_,j\_,k\_,l\_} := -p_{\_}\,x_{\_}\ /.\ \{p_{\_} \to p_i - p_k,\ x_{\_} \to x_i - x_k\}; \end{array}\right);$$

In[∘]:=
$$\text{CF}[Q_{i,j,k,l}]$$

Out[∘]=
$$p_i\,x_i - p_k\,x_i - p_i\,x_k + p_k\,x_k$$

In[∘]:=
$$\text{Collect}[Q_{i,j,k,l},\ p\_,\ \text{Simplify}]$$

Out[∘]=
$$p_i\,(x_i - x_k) + p_k\,(-x_i + x_k)$$

In[*]:= **Collect[$Q_{i,j,k,l}$, x_, Simplify]**

Out[*]=

$(p_i - p_k)\, x_i + (-p_i + p_k)\, x_k$

In[*]:= **CF[$Q_{i,j,k,l}$ /. {$p_i \to (p_+ + p_-)/2$, $p_k \to (p_+ - p_-)/2$, $x_i \to (x_+ + x_-)/2$, $x_k \to (x_+ - x_-)/2$}]**

Out[*]=

$p_- \, x_-$

In[*]:= **CF[$\overline{Q}_{i,j,k,l}$ /. {$p_i \to (p_+ + p_-)/2$, $p_k \to (p_+ - p_-)/2$, $x_i \to (x_+ + x_-)/2$, $x_k \to (x_+ - x_-)/2$}]**

Out[*]=

$-p_- \, x_-$

In[*]:= **CF[$Q_{i,j,k,l} - \overline{Q}_{i,j,k,l}$ /. {$p_i \to (p_+ + p_-)/2$, $p_k \to (p_+ - p_-)/2$, $x_i \to (x_+ + x_-)/2$, $x_k \to (x_+ - x_-)/2$}]**

Out[*]=

$2\, p_- \, x_-$

The Trefoil



In[*]:= **L = CF[$Q_{1,2,3,4} + Q_{3,2,5,4} + Q_{5,2,1,4}$]**

Out[*]=

$2\, p_1\, x_1 - p_3\, x_1 - p_5\, x_1 - p_1\, x_3 + 2\, p_3\, x_3 - p_5\, x_3 - p_1\, x_5 - p_3\, x_5 + 2\, p_5\, x_5$

In[*]:= **{CF[(∂_{p_1} L) + (∂_{p_3} L) + (∂_{p_5} L)], CF[(∂_{x_1} L) + (∂_{x_3} L) + (∂_{x_5} L)]}**

Out[*]=

{0, 0}

In[*]:= **$\int \mathbb{E}[L]\, d\{p_1, p_3, x_1, x_3\}$**

Out[*]=

$\frac{\mathbb{E}[0]}{3}$

In[*]:= **$\int \mathbb{E}[L]\, d\{p_1, p_5, x_1, x_5\}$**

Out[*]=

$\frac{\mathbb{E}[0]}{3}$

In[*]:= **CF[$Q_{1,2,3,4} + Q_{3,2,5,4} + Q_{5,2,1,4}$ /. {$p_1 \to -p_3 - p_5$, $x_1 \to -x_3 - x_5$}]**

Out[*]=

$6\, p_3\, x_3 + 3\, p_5\, x_3 + 3\, p_3\, x_5 + 6\, p_5\, x_5$

Re-sectioning

In[∘]:= $\{CF[Q_{i,j,k,l} /. (p \mid x)_i \to 0],\ CF[Q_{i,j,k,l} /. (p \mid x)_k \to 0]\}$

Out[∘]=

$\{p_k\, x_k,\ p_i\, x_i\}$

In[∘]:= $\{lhs, rhs\} = \left\{\int \mathbb{E}[Q_{i,j,k,l} /. (p \mid x)_i \to 0]\, d\{p_k, x_k\},\ \int \mathbb{E}[Q_{i,j,k,l} /. (p \mid x)_k \to 0]\, d\{p_i, x_i\}\right\}$

Out[∘]=

$\{-i\, \mathbb{E}[0],\ -i\, \mathbb{E}[0]\}$

In[∘]:= lhs == rhs

Out[∘]=

True

In[∘]:= $f = \xi_j\, x_j + \pi_j\, p_j + \xi_l\, x_l + \pi_l\, p_l;$

$\{lhs, rhs\} =$

$\left\{\int \mathbb{E}[(Q_{i,j,k,l} + f) /. (p \mid x)_i \to 0]\, d\{p_k, x_k\},\ \int \mathbb{E}[(Q_{i,j,k,l} + f) /. (p \mid x)_k \to 0]\, d\{p_i, x_i\}\right\}$

lhs == rhs

Out[∘]=

$\{-i\, \mathbb{E}[p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l],\ -i\, \mathbb{E}[p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l]\}$

Out[∘]=

True

In[∘]:= $f = \xi_j\, x_j + \pi_j\, p_j + \xi_l\, x_l + \pi_l\, p_l + \pi\, (p_i - p_k) + \xi\, (x_i - x_k);$

$\{lhs, rhs\} =$

$\left\{\int \mathbb{E}[(Q_{i,j,k,l} + f) /. (p \mid x)_i \to 0]\, d\{p_k, x_k\},\ \int \mathbb{E}[(Q_{i,j,k,l} + f) /. (p \mid x)_k \to 0]\, d\{p_i, x_i\}\right\}$

lhs == rhs

Out[∘]=

$\{-i\, \mathbb{E}[-\pi\, \xi + p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l],\ -i\, \mathbb{E}[-\pi\, \xi + p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l]\}$

Out[∘]=

True

In[∘]:= $f = \xi_j\, x_j + \pi_j\, p_j + \xi_l\, x_l + \pi_l\, p_l + \pi\, (p_i - p_k) + \xi\, (x_i - x_k);$

$\{lhs, rhs\} =$

$\left\{\int \mathbb{E}\left[\left(\overline{Q}_{i,j,k,l} + f\right) /. (p \mid x)_i \to 0\right] d\{p_k, x_k\},\ \int \mathbb{E}\left[\left(\overline{Q}_{i,j,k,l} + f\right) /. (p \mid x)_k \to 0\right] d\{p_i, x_i\}\right\}$

lhs == rhs

Out[∘]=

$\{-i\, \mathbb{E}[\pi\, \xi + p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l],\ -i\, \mathbb{E}[\pi\, \xi + p_j\, \pi_j + p_l\, \pi_l + x_j\, \xi_j + x_l\, \xi_l]\}$

Out[∘]=

True

R2b



In[∘]:= $Q_{1,2,3,4} + \overline{Q}_{3,2,5,4}$ // Expand

Out[∘]=

$$p_1 x_1 - p_3 x_1 - p_1 x_3 + p_5 x_3 + p_3 x_5 - p_5 x_5$$

In[∘]:= Expand@Collect$\left[Q_{1,2,3,4} + \overline{Q}_{3,2,5,4},\ (p \mid x)_3,\ F\right]$ /. F → Factor

Out[∘]=

$$p_1 x_1 + (-p_1 + p_5)\, x_3 - p_5 x_5 + p_3 (-x_1 + x_5)$$

In[∘]:= Expand$\left[Q_{1,2,3,4} + \overline{Q}_{3,2,5,4} \text{ /. } \{p_5 \to p_1,\ x_5 \to x_1\}\right]$

Out[∘]=

0

R3



In[∘]:= lhs = Simplify$[Q_{1,2,7,6} + Q_{2,3,4,7} + Q_{7,4,5,6}]$;
rhs = Simplify$[Q_{2,3,7,1} + Q_{1,7,5,6} + Q_{7,3,4,5}]$

Out[∘]=

$$(p_1 - p_5)(x_1 - x_5) + (p_2 - p_7)(x_2 - x_7) + (p_4 - p_7)(x_4 - x_7)$$

In[∘]:= Simplify[lhs == rhs]

Out[∘]=

$$p_4 x_2 + p_2 x_4 + p_7 (x_1 - x_2 - x_4 + x_5) + p_1 x_7 + p_5 (-x_1 + x_7) == p_1 x_5 + (p_2 + p_4)\, x_7$$

In[∘]:= Expand@Collect$[\text{lhs},\ (p \mid x)_7,\ F]$ /. F → Simplify

Out[∘]=

$$p_1 x_1 + (p_2 - p_4)(x_2 - x_4) + p_7 (-x_1 - x_5) + p_5 x_5 + (-p_1 - p_5)\, x_7 + 2 p_7 x_7$$

In[∘]:= $\int \mathbb{E}[\text{lhs}]\, d\{p_7, x_7\}$

Out[∘]=

$$-\frac{1}{2}\, i\, \mathbb{E}\left[\frac{p_1 x_1}{2} - \frac{p_5 x_1}{2} + p_2 x_2 - p_4 x_2 - p_2 x_4 + p_4 x_4 - \frac{p_1 x_5}{2} + \frac{p_5 x_5}{2}\right]$$

In[∘]:=
```
Expand@Collect[rhs, (p | x)_7, F] /. F → Simplify
```
Out[∘]=

$$p_2 x_2 + p_7 (-x_2 - x_4) + p_4 x_4 + (p_1 - p_5)(x_1 - x_5) + (-p_2 - p_4) x_7 + 2 p_7 x_7$$

In[∘]:=
```
Simplify[(∫𝔼[rhs] ⅆ{p_7, x_7}) == (∫𝔼[lhs] ⅆ{p_7, x_7})]
```
Out[∘]=

$$\mathbb{E}\left[\frac{1}{2}(p_2 - p_4)(x_2 - x_4) + p_1 (x_1 - x_5) + p_5 (-x_1 + x_5)\right] == \mathbb{E}\left[\frac{1}{2}(2(p_2 - p_4)(x_2 - x_4) + p_1 (x_1 - x_5) + p_5 (-x_1 + x_5))\right]$$

In[∘]:=
```
(∫𝔼[rhs] ⅆ{p_7, x_7}) ≡ (∫𝔼[lhs] ⅆ{p_7, x_7})
```
Out[∘]=

$$(p_2 - p_4)(x_2 - x_4) + p_5 (x_1 - x_5) + p_1 (-x_1 + x_5) == 0$$

R1s



In[∘]:=
```
R1s = Simplify@{Q_{2,3,2,1}, Q̄_{2,3,2,1}, Q̄_{1,2,1,3}, Q_{1,2,1,3}}
```
Out[∘]=

{0, 0, 0, 0}

R2C$_+$



In[∘]:=
```
lhs = Collect[Q̄_{4,1,2,5} + Q_{2,3,4,5}, t, Simplify]
```
Out[∘]=

0

R2C$_-$

In[∘]:= **lhs = Collect[$\overline{Q}_{2,5,4,1}$ + $Q_{4,5,2,3}$, t, Simplify]**

Out[∘]=

0